# $Id$
# Maintainer: Gaetan Bisson <bisson@archlinux.org>
# Contributor: dorphell <dorphell@archlinux.org>

pkgname=rcs
pkgver=5.9.4
pkgrel=2
pkgdesc='Revision Control System: manages multiple revisions of files'
url='http://www.gnu.org/software/rcs/'
license=('GPL3')
arch=('x86_64')
depends=('ed')
validpgpkeys=('748EA0E81CB8A7489BFA6CE4670322244C807502')
source=("ftp://ftp.gnu.org/gnu/${pkgname}/${pkgname}-${pkgver}.tar.xz"{,.sig} rcs-c++11.patch)
sha256sums=('063d5a0d7da1821754b80c639cdae2c82b535c8ff4131f75dc7bbf0cd63a5dff'
            'SKIP'
            '569a8d5c851d2619393474a681e05c2eb83912e71adcf1124b895478c91e1fbb')

prepare() {
  cd $pkgname-$pkgver
  patch -p1 -i ../rcs-c++11.patch # Fix build with C++11
}

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make
}

check() {
  cd ${pkgname}-${pkgver}
  make check
}

package() {
  cd ${pkgname}-${pkgver}
  make prefix="${pkgdir}/usr" install
}
